

REPEAT

Syntax 1

REPEAT

statement... $\left[\begin{array}{l} \{\text{UNTIL}\} \\ \{\text{WHILE}\} \end{array} \right] \text{logical-condition}$

END-REPEAT (*structured mode only*)

LOOP (*reporting mode only*)

Syntax 2

REPEAT

$\left[\begin{array}{l} \{\text{UNTIL}\} \\ \{\text{WHILE}\} \end{array} \right] \text{logical-condition}$ *statement...*

END-REPEAT (*structured mode only*)

LOOP (*reporting mode only*)

Related Statement: FOR

Function

The REPEAT statement is used to initiate a processing loop.

If no logical condition is specified, the loop must be exited by an ESCAPE, STOP or TERMINATE statement specified within the loop. If a logical condition is specified, the condition determines when the execution of the loop is to be terminated.

Using syntax 1, the statements are executed one or more times.

Using syntax 2, the statements are executed zero or more times.

The placement of the condition (either at the beginning or at the end of the loop) determines when it is to be evaluated.

For further information on logical conditions, see the section Logical Condition Criteria.

UNTIL

REPEAT

UNTIL

The processing loop will be continued until the logical condition becomes true.

WHILE

The processing loop will be continued as long as the logical condition is true.

Example 1

```
/* EXAMPLE 'RPTEX1S': REPEAT (STRUCTURED MODE)
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
1 #PERS-NR (A8)
END-DEFINE
*****
REPEAT
  INPUT 'ENTER A PERSONNEL NUMBER:' #PERS-NR
  IF #PERS-NR = ''
    ESCAPE BOTTOM
  END-IF
  FIND EMPLOY-VIEW WITH PERSONNEL-ID = #PERS-NR
  IF NO RECORD FOUND
    REINPUT 'NO RECORD FOUND'
  END-NOREC
  DISPLAY NOTITLE NAME
  END-FIND
END-REPEAT
*****
END
```

ENTER A PERSONNEL NUMBER: 11500304

NAME

KLUGE

Equivalent reporting-mode example: See the program RPTEX1R in the library SYSEXRM.

Example 2:

```
/* EXAMPLE 'RPTEX2S': REPEAT (WHILE AND UNTIL OPTIONS)
*****
DEFINE DATA LOCAL
1 #X (I1) INIT <0>
1 #Y (I1) INIT <0>
END-DEFINE
*****
REPEAT WHILE #X <= 5
    ADD 1 TO #X
    WRITE NOTITLE '=' #X
END-REPEAT
*****
SKIP 1
REPEAT
    ADD 1 TO #Y
    WRITE '=' #Y
    UNTIL #Y = 6
END-REPEAT
*****
END
```

```
#X:      1
#X:      2
#X:      3
#X:      4
#X:      5
#X:      6

#Y:      1
#Y:      2
#Y:      3
#Y:      4
#Y:      5
#Y:      6
```

Equivalent reporting-mode example: See the program RPTEX2R in the library SYSEXRM.